Eulerin yhtälön muuntaminen vakiokertoimiseksi

Mathematicaa voidaan käyttää yleisen Eulerin yhtälön muuntamiseen vakiokertoimiseksi yhtälöksi sijoituksella $x = e^t$. Seuraava esitys toimii periaatteessa mille tahansa kertaluvulle n. Jos $n > 20$, alkaa laskenta kuitenkin vaatia varsin paljon resursseja.

Syötetään Eulerin yhtälön kertaluku ja muodostetaan yhtälö:

```
n = 5
```

```
5
```

```
euleryht = Sum[a_k x^k D[y[x], {x, k}], {k, 0, n}] == 0
```

$$a_0\, y[x] + x\, a_1\, y'[x] + x^2\, a_2\, y''[x] + x^3\, a_3\, y^{(3)}[x] + x^4\, a_4\, y^{(4)}[x] + x^5\, a_5\, y^{(5)}[x] == 0$$

Sijoituksen seurauksena syntyy uusi tuntematon funktio $u(t) = y(e^t)$. Funktiot y ja u toisiinsa sitova yhtälö talletetaan nimelle `sijyhtalo`:

```
sijyhtalo = u[t] == y[Exp[t]]
```

$$u[t] == y[e^t]$$

Tätä yhtälöä derivoidaan muuttujan t suhteen n kertaa, jotta saadaan vastaavat derivaattojen väliset yhtälöt; nämä kerätään listaksi:

```
yhtalot = Table[D[sijyhtalo, {t, k}], {k, 0, n}];
yhtalot // TableForm
```

$$u[t] == y[e^t]$$
$$u'[t] == e^t\, y'[e^t]$$
$$u''[t] == e^t\, y'[e^t] + e^{2t}\, y''[e^t]$$
$$u^{(3)}[t] == e^t\, y'[e^t] + 3\, e^{2t}\, y''[e^t] + e^{3t}\, y^{(3)}[e^t]$$
$$u^{(4)}[t] == e^t\, y'[e^t] + 7\, e^{2t}\, y''[e^t] + 6\, e^{3t}\, y^{(3)}[e^t] + e^{4t}\, y^{(4)}[e^t]$$
$$u^{(5)}[t] == e^t\, y'[e^t] + 15\, e^{2t}\, y''[e^t] + 25\, e^{3t}\, y^{(3)}[e^t] + 10\, e^{4t}\, y^{(4)}[e^t] + e^{5t}\, y^{(5)}[e^t]$$

Tuntemattomat, ts. funktion y derivaatat kerätään omaksi listakseen, jossa muuttujaksi otetaan t:

```
tuntemattomat = Table[D[y[x], {x, k}], {k, 0, n}] /. x -> Exp[t]
```

$$\{y[e^t],\, y'[e^t],\, y''[e^t],\, y^{(3)}[e^t],\, y^{(4)}[e^t],\, y^{(5)}[e^t]\}$$

Jotta alkuperäinen differentiaaliyhtälö saadaan muunnetuksi funktiota u koskevaksi, yhtälöryhmästä ratkaistaan funktion y derivaatat ja sijoitetaan nämä differentiaaliyhtälöön; muuttujaksi otetaan t:

```
sij = Solve[yhtalot, tuntemattomat]
```

$\{\{y^{(5)}[e^t] \to e^{-5t}(24\,u'[t] - 50\,u''[t] + 35\,u^{(3)}[t] - 10\,u^{(4)}[t] + u^{(5)}[t]),$
$y[e^t] \to u[t],\ y^{(4)}[e^t] \to -e^{-4t}(6\,u'[t] - 11\,u''[t] + 6\,u^{(3)}[t] - u^{(4)}[t]),$
$y^{(3)}[e^t] \to e^{-3t}(2\,u'[t] - 3\,u''[t] + u^{(3)}[t]),\ y''[e^t] \to -e^{-2t}(u'[t] - u''[t]),\ y'[e^t] \to e^{-t}\,u'[t]\}\}$

```
vakiokertyht = euleryht /. x -> Exp[t] /. First[sij]
```

$a_0\,u[t] + a_1\,u'[t] - a_2\,(u'[t] - u''[t]) +$
$a_3\,(2\,u'[t] - 3\,u''[t] + u^{(3)}[t]) - a_4\,(6\,u'[t] - 11\,u''[t] + 6\,u^{(3)}[t] - u^{(4)}[t]) +$
$a_5\,(24\,u'[t] - 50\,u''[t] + 35\,u^{(3)}[t] - 10\,u^{(4)}[t] + u^{(5)}[t]) == 0$

Tällöin on saatu Eulerin yhtälöä vastaava vakiokertoiminen yhtälö. Jotta tämä hahmottuisi selkeämmin, termit kootaan funktion *u* derivaattojen mukaan ryhmiteltyinä:

```
derivaatat = Table[D[u[t], {t, k}], {k, 0, n}]
```

$\{u[t],\ u'[t],\ u''[t],\ u^{(3)}[t],\ u^{(4)}[t],\ u^{(5)}[t]\}$

```
ryhmitettyvakiokertyht = Collect[vakiokertyht[[1]], derivaatat] == 0
```

$a_0\,u[t] + (a_1 - a_2 + 2\,a_3 - 6\,a_4 + 24\,a_5)\,u'[t] + (a_2 - 3\,a_3 + 11\,a_4 - 50\,a_5)\,u''[t] +$
$(a_3 - 6\,a_4 + 35\,a_5)\,u^{(3)}[t] + (a_4 - 10\,a_5)\,u^{(4)}[t] + a_5\,u^{(5)}[t] == 0$

`Collect`-funktiota ei voida kohdistaa suoraan yhtälöön, vaan ainoastaan sen vasempaan puoleen `vakiokertyht[[1]]`, mikä tuo syötteisiin ylimääräistä monimutkaisuutta. Toisaalta kaikki syötteet voidaan pakata myös yhdelle riville ja muutoinkin hyödyntää Mathematican ohjelmointikielen piirteitä:

```
Function[x, Collect[x, Table[D[u[t], {t, k}], {k, 0, n}]]] /@ vakiokertyht
```

$a_0\,u[t] + (a_1 - a_2 + 2\,a_3 - 6\,a_4 + 24\,a_5)\,u'[t] + (a_2 - 3\,a_3 + 11\,a_4 - 50\,a_5)\,u''[t] +$
$(a_3 - 6\,a_4 + 35\,a_5)\,u^{(3)}[t] + (a_4 - 10\,a_5)\,u^{(4)}[t] + a_5\,u^{(5)}[t] == 0$

Saadun yhtälön vasemmasta puolesta voidaan myös poimia eri derivaattojen kertoimet ja esittää nämä taulukkona:

```
Table[Coefficient[ryhmitettyvakiokertyht[[1]], D[u[t], {t, k}]], {k, 0, n}] // TableForm
```

a_0
$a_1 - a_2 + 2\,a_3 - 6\,a_4 + 24\,a_5$
$a_2 - 3\,a_3 + 11\,a_4 - 50\,a_5$
$a_3 - 6\,a_4 + 35\,a_5$
$a_4 - 10\,a_5$
a_5

Kertoimet ovat todellakin vakioita.

Linkit

Eulerin yhtälön muuntaminen vakiokertoimiseksi ja ratkaiseminen
yhtälön muuntaminen sijoituksella

SKK 30.04.2001